AMORT – Loan Amortization Schedule

Abstract

AMORT is a program written in 1979 for the HP-34C programmable calculator to generate either a complete amortization schedule showing each period of a fully amortized loan or a partial schedule between two given periods. One worked example is included.

Keywords: *amortization, loan, schedule, partial, programmable calculator, RPN, HP-34C, HP-92 Investor, financial calculator*

1. Introduction

AMORT is a short *(70 steps) RPN* program that I wrote in 1979 for the *HP-34C* calculator (will also run *as-is* or with minor modifications in many *RPN* models), which generates a complete amortization schedule showing each period in a fully amortized loan, displaying the amount paid in interest, paid to principal, and the remaining balance. Last, it displays the remaining balance on the loan and the total amounts paid to principal and interest.

It can also display a partial schedule between two given periods or even for just one. It essentially duplicates the *Loan Amortization Schedule* functionality of *Hewlett-Packard's HP-92 Investor* financial desktop calculator.

2. Program Listing

```
01 ♦LBL A     15 STO 0      29 PSE        43 R↑         57 y^x        - 70 steps
02  STO 3     16 STO 6      30 FIX 2      44 PSE        58 STO 5      - uses registers R0-R6, RI
03  R↓        17 ♦LBL 0     31 RCL 2      45 ISG        59  1
04  STO 2     18  RCL I     32 R↑         46 GTO 0 ▸    60 STO- 4     - all PSE instructions may be
05  R↓        19  INT       33 R↑         47 RCL 6      61  -           replaced by R/S instructions
06  STO 1     20  STO 4     34 PSE        48 PSE        62 RCL 1        or print statements. See Note 2
07  RTN       21 GSB 1 ▸    35 STO+ 6     49 RCL 0      63  ÷
08 ♦LBL B     22  ENTER     36 X↔Y        50 RTN        64 RCL 2      - the symbols ♦ and ▸ are purely
09  EEX       23 GSB 1 ▸    37 R↓         51 ♦LBL 1     65  x           cosmetic, to indicate branching
10   3        24   -        38  -         52 RCL 1      66 RCL 3
11   ÷        25 RCL 2      39 PSE        53  1         67  +
12   +        26   +        40 STO+ 0     54  +         68 RCL 5
13  STO I     27 RCL I      41 X↔Y        55 RCL 4      69  ÷
14   0        28 FIX 0      42 LSTX       56 CHS        70 RTN
```

3. Usage Instructions

1) Input **i%** *(annual interest rate)*, **PMT** *(annual mortgage payment)* and **PV** *(amount of loan)*:

i% **ENTER↑** **PMT** **ENTER↑** **PV** **A** ***i%***

2) Input **P1** *(1st period of the schedule)* and **P2** *(last period of schedule)* and start the amortization schedule:

P1 **ENTER↑** **P2** **B** **...** the amortization schedule starts and outputs the following:

For each period $\mathbf{P_k}$ from **P1** to **P2**:

$\boldsymbol{P_k}$	*Period*
INT	*Interest amount*
PRN	*Amount paid to principal*
BAL	*Remaining balance*

Finally, it outputs:

$\boldsymbol{\Sigma}$ ***INT***	*Total amount paid to principal*
$\boldsymbol{\Sigma}$ ***PRN***	*Total amount paid to interest*

Notes: - all values must be positive; the interest rate **i%** must be entered thus: ***9%*** as ***0.09***, ***148%*** as ***1.48***, etc.
- for a *single period*, just input **P2** = **P1** (the totals ***Σ INT*** and ***Σ PRN*** are then redundant).
- to produce the schedule for other periods, there's no need to re-input **i%**, **PMT** and **PV** if they don't change because they're kept unaltered by the program. Else, repeat *step 1* above to input the new values.

4. Examples

The following example, adapted from a typical *HP-92 Investor* brochure, can be useful to check that the program is correctly entered and to better understand its usage.

4.1 Example 1

An investor receives a loan of $100,000 for 20 years at 9% annual interest, which results in an annual mortgage payment of $10,954.65 . Generate an amortization schedule for the first 3 years, then for the 15th year.

0.09 ***(i%)*** **ENTER↑** 10954.65 ***(PMT)*** **ENTER↑** 100000 ***(PV)*** **A** 0.09 ***(i%)***

1 ***(P1)*** **ENTER↑** 3 ***(P2)*** **B**

1	*P*	***Period 1***
9000.00	*INT*	*Interest amount*
1954.65	*PRN*	*Principal amount*
98045.35	*BAL*	*Remaining balance*
2	*P*	***Period 2***
8824.08	*INT*	*Interest amount*
2130.57	*PRN*	*Principal amount*
95914.78	*BAL*	*Remaining balance*
3	*P*	***Period 3***
8632.33	*INT*	*Interest amount*
2322.32	*PRN*	*Principal amount*
93592.46	*BAL*	*Remaining balance*
26456.41	*Σ INT*	*Total amount paid to interest*
6407.54	*Σ PRN*	*Total amount paid to principal*

15 ***(P1)*** **ENTER↑** ***(P2 = P1)*** **B**

15	*P*	***Period 15***
4422.74	*INT*	*Interest amount*
6531.91	*PRN*	*Principal amount*
42609.69	*BAL*	*Remaining balance*
4422.74	*Σ INT*	*Total amount paid to interest (redundant)*
6531.91	*Σ PRN*	*Total amount paid to principal (redundant)*

Notes

1. This program essentially duplicates the *HP-92 Investor*'s **AMORT** functionality, albeit perhaps less accurately.

2. The **PSE** instructions at steps *29*, *34*, *39*, *44*, *48* may be duplicated to make the output stay longer on the display or better still, they might be replaced by **R/S** to make the program stop at each output and have ample time to write it down, then simply press **R/S** to continue. If using a printing model, the **PSE** instructions might be replaced by printing statements.

References

Hewlett-Packard (1988). *HP-92 Investor financial desktop calculator brochure*.

Copyrights